Supporting Information

‘Determination of Size and Concentration of Gold Nanoparticles from UV/vis Spectra’

*Wolfgang Haiss**[1,2], Nguyen T. K. Thanh[1,2], Jenny Aveyard[1] and David G. Fernig[2]*

Centre for Nanoscale Science,
[1]Department of Chemistry, University of Liverpool, Crown Street L69 7ZD, U.K
[2]School of Biological Sciences, University of Liverpool, Crown Street L69 7ZB, U.K

Table of Content

1. Tabular material for the determination of the size of uncoated spherical gold nanoparticles in water from UV/vis spectra.

The values in Tables S-1 and S-2 can be used to calculate the size of GNPs form UV/vis spectra. For particles having a diameter larger than 35 nm Table S-1 and Table S-2 can be used to determine the particle diameter *d*. For smaller particles the absorbance ratio of Table S-2 can be used but it should be noted that particle sizes can be obtained from Equation 12 with slightly improved precision if the initial gold concentration is known. For particle diameters smaller than 10 nm the values in Table S-2 should only be considered as an estimate since surface effects may get increasingly important in this size region.

λ_{spr} / nm	d / nm	λ_{spr} / nm	d / nm	λ_{spr} / nm	d / nm
525.0	32	534.9	58	552.1	84
525.6	34	535.9	60	553.8	86
526.2	36	536.9	62	555.7	88
526.8	38	538.0	64	557.6	90
527.5	40	539.2	66	559.6	92
528.2	42	540.4	68	561.7	94
528.9	44	541.6	70	563.9	96
529.6	46	542.9	72	566.2	98
530.4	48	544.3	74	568.6	100
531.2	50	545.7	76	571.1	102
532.1	52	547.2	78	573.7	104
533.0	54	548.8	80	576.5	106
533.9	56	550.4	82	579.3	108

Table S-1: Position of the surface plasmon resonance peak in dependence of the particle diameter calculated from Equation 10.

A_{spr}/A_{450}	d / nm	A_{spr}/A_{450}	d / nm	A_{spr}/A_{450}	d / nm
1.10	3	1.56	12	1.96	40
1.19	4	1.61	14	2.00	45
1.27	5	1.65	16	2.03	50
1.33	6	1.69	18	2.07	55
1.38	7	1.73	20	2.10	60
1.42	8	1.80	25	2.12	65
1.46	9	1.86	30	2.15	70
1.50	10	1.92	35	2.17	75

Table S-2: Ratio of the absorbance of GNPs at the surface plasma resonance peak (A_{spr}) to the absorbance at 450 nm (A_{450}) in dependence the particle diameter calculated from Equation 11 using the fit parameters to the experimental results.

2. Tabular material for the determination the concentration of uncoated spherical gold nanoparticles in water from UV/vis spectra.

The data of $\varepsilon_{450}(d)$ in Table S-3 can be used to calculate the particle concentration c in mol per litre from the absorption A at 450 nm for a standard path length l of 1 cm according to:

$$c = A_{450}/\varepsilon_{450}$$

The data have been experimentally verified in the d range from 5 – 100 nm (see Figure 8). For particle diameters smaller than 5 nm the values should only be considered as an estimate since surface effects may get increasingly important in this size region.

d/ nm	ε_{450} / $M^{-1}cm^{-1}$	*d*/ nm	ε_{450} / $M^{-1}cm^{-1}$	*d*/ nm	ε_{450} / $M^{-1}cm^{-1}$
2	4.25E+05	35	3.21E+09	68	2.50E+10
3	1.49E+06	36	3.52E+09	69	2.61E+10
4	3.62E+06	37	3.84E+09	70	2.71E+10
5	7.20E+06	38	4.18E+09	71	2.82E+10
6	1.26E+07	39	4.54E+09	72	2.93E+10
7	2.03E+07	40	4.92E+09	73	3.05E+10
8	3.07E+07	41	5.32E+09	74	3.16E+10
9	4.43E+07	42	5.74E+09	75	3.28E+10
10	6.15E+07	43	6.18E+09	76	3.40E+10
11	8.27E+07	44	6.65E+09	77	3.52E+10
12	1.09E+08	45	7.13E+09	78	3.64E+10
13	1.39E+08	46	7.65E+09	79	3.77E+10
14	1.76E+08	47	8.18E+09	80	3.89E+10
15	2.18E+08	48	8.74E+09	81	4.02E+10
16	2.67E+08	49	9.32E+09	82	4.14E+10
17	3.24E+08	50	9.92E+09	83	4.27E+10
18	3.87E+08	51	1.06E+10	84	4.40E+10
19	4.60E+08	52	1.12E+10	85	4.53E+10
20	5.41E+08	53	1.19E+10	86	4.65E+10
21	6.31E+08	54	1.26E+10	87	4.78E+10
22	7.31E+08	55	1.33E+10	88	4.91E+10
23	8.42E+08	56	1.41E+10	89	5.04E+10
24	9.64E+08	57	1.48E+10	90	5.17E+10
25	1.10E+09	58	1.57E+10	91	5.30E+10
26	1.24E+09	59	1.65E+10	92	5.43E+10
27	1.40E+09	60	1.73E+10	93	5.56E+10
28	1.58E+09	61	1.82E+10	94	5.69E+10
29	1.76E+09	62	1.91E+10	95	5.82E+10
30	1.96E+09	63	2.00E+10	96	5.94E+10
31	2.18E+09	64	2.10E+10	97	6.07E+10
32	2.41E+09	65	2.19E+10	98	6.19E+10
33	2.66E+09	66	2.29E+10	99	6.31E+10
34	2.93E+09	67	2.40E+10	100	6.44E+10

Table S-3: Molar decadic extinction coefficient (ε) at λ = 450 nm calculated from the fit (Equation 13) to the theoretical extinction efficiencies for gold nano particles in water with diameter d ranging from 2 to 100 nm.

3. The FORTRAN 77 program 'mie-mfp.f' which was used to calculate $Q_{ext}(\lambda,d)$.

```
      PROGRAM MIE-MFP
      REAL A(1100), B(1100), C(1100)
      COMPLEX RFREL1,RFREL2
      CHARACTER*64 file_in, file_out

C     ****************************************************************
C     This program calculates the scattering efficiency (QSCA;col2),
C     the extinction efficiency (QEXT;col3) for a coated sphere
C     using Mie theory. The wavelength of the incident light
C     is scanned between two values with a step-width.
C     The complex refractive index of the sphere as a function
C     of the wavelength/nm in vacuum is read in from a separate file.
C     Linear interpolation is used to interpolate these
C     experimentally determined data.
C     ****************************************************************

C     ****************************************************************
C     Input of parameters:
      WRITE(6,*) 'ENTER THE START WAVELENGT / NM:'
      READ(5,*) WAVEL1
      WRITE(6,*) 'ENTER END WAVELENGT / NM:'
      READ(5,*) WAVEL2
      WRITE(6,*) 'ENTER THE NUMBER OF VALUES TO BE CALCULATED:'
      READ(5,*) NUMVAL
      WRITE(6,*) 'REFRACTIVE INDEX OF THE SURROUNDING MEDIUM:'
      READ(5,*) REFMED
      WRITE(6,*) 'REAL PART REFRACTIVE INDEX OF THE COATING:'
      READ(5,*) REFRE2
      WRITE(6,*) 'IMMAG. PART OF THE REFR. INDEX OF THE COATING:'
      READ(5,*) REFIM2
      WRITE(6,*) 'SPHERE RADIUS / NM:'
      READ(5,*) RADCOR
      WRITE(6,*) 'COATING RADIUS / NM:'
      READ(5,*) RADCOT
      WRITE(6,*) 'ENTER THE PLASMA FREQUENCY IN HZ/1E+14 (138. for Au):'
      READ(5,*) OMP
      WRITE(6,*) 'ENTER THE COLLISION FREQ. IN Hz/1E+14 (0.333 for Au):'
      READ(5,*) OM0
      WRITE(6,*) 'ENTER END FERMI VELOCITY IN cm/s (1.4E+8 for Au):'
      READ(5,*) FV
      WRITE(6,*) 'SPECIFY THE FILENAME OF THE CORE REFRACTIVE INDEX:'
      READ(5,'(A)') file_in
      WRITE(6,*) 'SPECIFY THE FILENAME OF THE OUTPUT FILE:'
      READ(5,'(A)') file_out
C     ****************************************************************

      pi=3.14159265
      TAU= (1/OM0)*1E-14
      FMPINF=FV*TAU

C     ****************************************************************
C     Output of parameters to file_out:
      OPEN(UNIT=11,FILE=file_out,STATUS='NEW')
      WRITE(11,*) 'WAV   ','QSCA   ','QEXT   ','NSTOP   '
C     ****************************************************************

C     ****************************************************************
C     Data of REFRE1 (col2;B(K)) and REFIM1 (col3;C(K))are read from
C     the input file into 1-dimensional arrays (A=WAVEL; B=n; C=k).
C     Note that the data-file must contain increasing values of the
C     wavelength in column 1 it also must contain a single number in
C     line 1, specifying the total Nr of Data lines.

      OPEN(UNIT=10,FILE=file_in,STATUS='OLD')
      READ(10,*) N
      DO 5 K=1,N
         READ(10,*) A(K), B(K), C(K)
 5    CONTINUE
C     ****************************************************************

C     Calculating the wavelength step-width:
      DELTA = (WAVEL2 - WAVEL1)/(NUMVAL - 1)
```

```
C     Repeating the calculations:
      DO 100 I = 1,NUMVAL
      WAVEL = WAVEL1 + (I-1)*DELTA

C     ***************************************************************
C     Linear interpolation of n and k from input file:
       IF(WAVEL.EQ.A(1)) THEN
            K=1
       ELSE
            K=2
 15         IF (A(K).LT.WAVEL) THEN
              K=K+1
              GO TO 15
            END IF
      rn = B(K-1) + (WAVEL - A(K-1))/(A(K) - A(K-1))*(B(K) - B(K-1))
      rk = C(K-1) + (WAVEL - A(K-1))/(A(K) - A(K-1))*(C(K) - C(K-1))
      END IF
C     ***************************************************************

      CALL MFP (FV,WAVEL,RADCOR,OMP,OM0,rn,rk,rnr,rkr,OM0R)
C     This Subroutine corrects the bulk values of n and k for small
C     particles taking into account the 'mean free path effect' on
C     the 'free' conduction electrons in the Drude model.

      REFRE1 = rnr
      REFIM1 = rkr

      TAUR=(1/OM0R)*1E-14
      FMPR=FV*TAUR



C     Calculating relative size parameters for core (X) and coating (Y)

      X=2.*pi*RADCOR*REFMED/WAVEL
      y=2.*pi*RADCOT*REFMED/WAVEL

C     Calculating the complex relative refractive indices:

      RFREL1=CMPLX(REFRE1,REFIM1)/REFMED
      RFREL2=CMPLX(REFRE2,REFIM2)/REFMED

      CALL BHCOAT (X,Y,RFREL1,RFREL2,QEXT,QSCA,QBACK,NSTOP)

C     Recalibration of the scattering efficiencies Q
C     with respect to RADCOR instead of RADCOT

         QEXT = QEXT*(RADCOT/RADCOR)*(RADCOT/RADCOR)
         QSCA = QSCA*(RADCOT/RADCOR)*(RADCOT/RADCOR)
         QBACK= QBACK*(RADCOT/RADCOR)*(RADCOT/RADCOR)

      WRITE (11,*) WAVEL,QSCA,QEXT,NSTOP

 100  CONTINUE


      WRITE(11,*) 'FMPINF:   ',FMPINF ,'cm '
      WRITE(11,*) 'FMPR:     ',FMPR ,'cm '
      WRITE(11,*) 'TAU:      ',TAU ,'s '
      WRITE(11,*) 'TAUR:     ',TAUR ,'s '
      WRITE(11,*) 'OMP:      ',OMP ,'1E+14Hz '
      WRITE(11,*) 'OM0:      ',OM0 ,'1E+14Hz '
      WRITE(11,*) 'FER-VEL:  ',FV,'cm/s '
      WRITE(11,*) 'REFMED:   ',REFMED
      WRITE(11,*) 'REFRE2:   ',REFRE2
      WRITE(11,*) 'REFIM2:   ',REFIM2
      WRITE(11,*) 'RADCOR:   ',RADCOR
      WRITE(11,*) 'RADCOT:   ',RADCOT
      WRITE(11,*) 'FILEIN:   ',file_in
      WRITE(11,*) 'FILEOUT:  ',file_out

      STOP
      END
```

```
      SUBROUTINE BHCOAT (X,Y,RFREL1,RFREL2,QEXT,QSCA,QBACK,NSTOP)
C     ************************************************************
C     The Subroutine BHCOAT was produced from the original FORTRAN
C     code of Bohren and Huffman which can be found in the appendix
C     of their Book: Bohren, C. F.; Huffman, D. R.
C     Absorption and Scattering of Light by Small Particles;
C     Wiley Interscience: New York, 1983
C     ************************************************************

      COMPLEX RFREL1,RFREL2,X1,X2,Y2,REFREL
      COMPLEX D1X1,D0X1,D1X2,D0X2,D1Y2,D0Y2
      COMPLEX XI0Y,XI1Y,XIY,CHI0Y2,CHI1Y2,CHIY2,CHI0X2,CHI1X2,CHIX2
      COMPLEX CHIPX2,CHIPY2,ANCAP,BNCAP,DNBAR,GNBAR,AN,BN,CRACK,BRACK
      COMPLEX XBACK,AMESS1,AMESS2,AMESS3,AMESS4
      DEL=1.0E-8
      X1=RFREL1*X
      X2=RFREL2*X
      Y2=RFREL2*Y
      YSTOP = Y + 4.*Y**.3333 + 2.
      REFREL=RFREL2/RFREL1
      NSTOP = YSTOP
      D0X1=CCOS(X1)/CSIN(X1)
      D0X2=CCOS(X2)/CSIN(X2)
      D0Y2=CCOS(Y2)/CSIN(Y2)
      PSI0Y=COS(Y)
      PSI1Y=SIN(Y)
      CHI0Y=-SIN(Y)
      CHI1Y=COS(Y)
      XI0Y=CMPLX(PSI0Y,-CHI0Y)
      XI1Y=CMPLX(PSI1Y,-CHI1Y)
      CHI0Y2=-CSIN(Y2)
      CHI1Y2=CCOS(Y2)
      CHI0X2=-CSIN(X2)
      CHI1X2=CCOS(X2)
      QSCA=0.0
      QEXT=0.0
      XBACK=CMPLX(0.0,0.0)
      N=1
      IFLAG=0
  200 RN=N
      PSIY=(2.*RN-1.)*PSI1Y/Y-PSI0Y
      CHIY=(2.*RN-1.)*CHI1Y/Y-CHI0Y
      XIY=CMPLX(PSIY,-CHIY)
      D1Y2=1./(RN/Y2-D0Y2)-RN/Y2
      IF (IFLAG.EQ.1) GO TO 999
      D1X1=1./(RN/X1-D0X1)-RN/X1
      D1X2=1./(RN/X2-D0X2)-RN/X2
      CHIX2=(2.*RN-1.)*CHI1X2/X2-CHI0X2
      CHIY2=(2.*RN-1.)*CHI1Y2/Y2-CHI0Y2
      CHIPX2=CHI1X2-RN*CHIX2/X2
      CHIPY2=CHI1Y2-RN*CHIY2/Y2
      ANCAP=REFREL*D1X1-D1X2
      ANCAP=ANCAP/(REFREL*D1X1*CHIX2-CHIPX2)
      ANCAP=ANCAP/(CHIX2*D1X2-CHIPX2)
      BRACK=ANCAP*(CHIY2*D1Y2-CHIPY2)
      BNCAP=REFREL*D1X2-D1X1
      BNCAP=BNCAP/(REFREL*CHIPX2-D1X1*CHIX2)
      BNCAP=BNCAP/(CHIX2*D1X2-CHIPX2)
      CRACK=BNCAP*(CHIY2*D1Y2-CHIPY2)
      AMESS1=BRACK*CHIPY2
      AMESS2=BRACK*CHIY2
      AMESS3=CRACK*CHIPY2
      AMESS4=CRACK*CHIY2
      IF(CABS(AMESS1).GT.DEL*CABS(D1Y2)) GO TO 999
      IF(CABS(AMESS2).GT.DEL) GO TO 999
      IF(CABS(AMESS3).GT.DEL*CABS(D1Y2)) GO TO 999
      IF(CABS(AMESS4).GT.DEL) GO TO 999
      BRACK=CMPLX(0.0,0.0)
      CRACK=CMPLX(0.0,0.0)
      IFLAG=1
  999 DNBAR=D1Y2-BRACK*CHIPY2
      DNBAR=DNBAR/(1.-BRACK*CHIY2)
      GNBAR=D1Y2-CRACK*CHIPY2
      GNBAR=GNBAR/(1.-CRACK*CHIY2)
      AN=(DNBAR/RFREL2+RN/Y)*PSIY-PSI1Y
      AN=AN/((DNBAR/RFREL2+RN/Y)*XIY-XI1Y)
      BN=(RFREL2*GNBAR+RN/Y)*PSIY-PSI1Y
```

```
      BN=BN/((RFREL2*GNBAR+RN/Y)*XIY-XI1Y)
      QSCA=QSCA+(2.*RN+1.)*(CABS(AN)*CABS(AN)+CABS(BN)*CABS(BN))
      XBACK=XBACK+(2.*RN+1.)*(-1.)**N*(AN-BN)
      QEXT=QEXT+(2.*RN+1.)*(REAL(AN)+REAL(BN))
      PSI0Y=PSI1Y
      PSI1Y=PSIY
      CHI0Y=CHI1Y
      CHI1Y=CHIY
      XI1Y=CMPLX(PSI1Y,-CHI1Y)
      CHI0X2=CHI1X2
      CHI1X2=CHIX2
      CHI0Y2=CHI1Y2
      CHI1Y2=CHIY2
      D0X1=D1X1
      D0X2=D1X2
      D0Y2=D1Y2
      N=N+1
      IF(N-1-NSTOP) 200,300,300
  300 QSCA=(2./(Y*Y))*QSCA
      QEXT=(2./(Y*Y))*QEXT
      QBACK=XBACK*CONJG(XBACK)
      QBACK=(1./(Y*Y))*QBACK
      RETURN
      END


      SUBROUTINE MFP (FV,WAVEL,RADCOR,OMP,OM0,rn,rk,rnr,rkr,OM0R)
C     ****************************************************************
C     This program corrects the bulk values of n and k for small
C     particles taking into account the 'mean free path effect' on
C     the 'free' conduction electrons in the extended Drude model.
C     All calculations performed in cm-g-s units.
C     Plasma Frequency and collision frequency both internally devided
C     by 1E+14 to avoid overflow
C     ****************************************************************
      CL = 2.998E+10
      pi = 3.14159265
C     Calculate EPS1 and EPS2 from rn and rk:
      EPS1 = rn*rn - rk*rk
      EPS2 = 2.*rn*rk
C     Calculate OM and A1 and A2:
      OM = (2.*pi*CL/(WAVEL*1.E-7))/1.E+14
      A1 = 1.-(OMP*OMP/(OM*OM + OM0*OM0))
      A2 = OMP*OMP*OM0/(OM*(OM*OM + OM0*OM0))
C     Contribution of the bond electrons to n (B1) and k (B2):
      B1 = EPS1 - A1
      B2 = EPS2 - A2
C     Calculate R dependent OM0 (OM0R)
      OM0R = OM0 + (FV/(RADCOR*1.E-7))/1.E+14
C     Calculate R dependent contributions of the free electrons:
      A1R = 1.-(OMP*OMP/(OM*OM + OM0R*OM0R))
      A2R = OMP*OMP*OM0R/(OM*(OM*OM + OM0R*OM0R))
C     Calculate R dependent EPS (EPS1R and EPS2R)
      EPS1R = A1R + B1
      EPS2R = A2R + B2
C     Reconvert EPS1R and EPS2R back to n and k:
      rnr = SQRT((A1R + B1)/2. + SQRT((A1R/2.+B1/2.)*(A1R/2.+B1/2.)
     _+(A2R/2.+B2/2.)*(A2R/2.+B2/2.)))
      rkr = SQRT(-(A1R + B1)/2. + SQRT((A1R/2.+B1/2.)*(A1R/2.+B1/2.)
     _+(A2R/2.+B2/2.)*(A2R/2.+B2/2.)))
C     WRITE(6,*) 'rn=         ',rn
C     WRITE(6,*) 'rnr=        ',rnr
C     WRITE(6,*) 'OM0=        ',OM0
C     WRITE(6,*) 'OM0R=               ',OM0R
C     WRITE(6,*) 'OM=          ',OM

      END
```